
Was ist Nagios?

● sowohl abruf als auch ereignisbasiertes 
Ueberwachungssystem

● aktive Ueberwachung von Netzwerkdiensten
● Ueberwachung lokaler Dienste, Prozesse, 

Hardware, Systemstati etc.
– aktiv ueber Agenten (nrpe, snmpd, nsclient++, ...)

– aktiv per Login (ssh, rsh, ...)

– optional passiv ueber Agenten und/oder 
Scheduler (z.B. crond und nsca, SNMPTraps)

Was ist Nagios?

● grosse Auswahl fertiger UeberwachungsPlugins 
(Checks) aus offiziellen und inoffiziellen Quellen

● einfaches Interface zum Erstellen eigener 
UeberwachungsPlugins

● ausserdem zahlreiche Erweiterungen, z.B. fuer
– das Empfangen und Auswerten passiv 

bereitgestellter Daten

– das Sammeln von PerformanceDaten

– das Erzeugen von Graphen/Trends

Screenshot: Nagios 3.0

Was ist Icinga?

● NagiosFork
– wurde im Mai 2009 von einer Gruppe (mit Nagios 

unzufriedener) ExNagiosEntwickler ins Leben 
gerufen

– soll schnelleren Entwicklungszyklus als Nagios 
bieten

– derzeit primaer optische Veraenderungen und 
Verbesserungen “unter der Haube”

● vollstaendig kompatibel zu Nagios, insbesondere 
zu NagiosUeberwachungsPlugins

Screenshot: Icinga 1.0

Screenshot: Icinga Alpha

Ueberwachungs-Plugins erstellen

● einfache Programme, beliebige Sprache
● Parameteruebernahme per Befehlszeile

– Hostname bzw. IPAdresse, Schwellwerte etc.

● Statusmeldung per ExitCode
– 0: OK, 1: WARNING, 2: CRITICAL, 3: UNKNOWN

● Mitteilungen ueber genau eine Ausgabezeile
– immer etwas auf STDOUT ausgeben und mit \n 

abschliessen

– uebliche Beispielausgabe: “OK : Works just fine”

Ueberwachungs-Plugins erstellen

● StandardBefehlszeilenparameter verwenden:
– h Hilfeseite des Plugins

– H Hostname/IPAdresse/UnixSocket

– I explizit IPAdresse

– w Schwelle fuer Warning

– c Schwelle fuer Critical

– u Username

– p Passwort oder Port

– V PluginVersion



Seite 10

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 8. Februar 2010

ePN: embedded Perl Nagios

● eingebetteter PerlInterpreter
– vergleichbar mit mod_perl fuer Apache

● fuehrt PerlProgramme direkt im Prozess aus
– UeberwachungsPlugins (Checks)

– EreignisHandler

– Benachrichtigungsprogramme

● reduziert Overhead von PerlProgrammen enorm
– keine Forks zur externen Programmausfuehrung

– kein Aufstarten des PerlInterpreters

ePN: embedded Perl Nagios

● Cache fuer PerlBytecode
– beschleunigt Start durch seltene Neukompilierung

– groessere Speicherbelegung gegenueber 
“normaler” PluginAusfuehrung

– Neustart von Nagios notwendig, um PerlModule 
neu zu laden

ePN: embedded Perl Nagios

● eventuell Anpassungen fuer ePN notwendig
– use strict; use warnings;

– Getopt::Std oder Getopt::Long benutzen

– keine Variablen in globalem Scope beschreiben

– Thread Safety beachten

– DATAHandle (__DATA__) funktioniert nicht

– BEGIN{} und END{} funktionieren nicht

– Handles und Sockets immer schliessen

– Timeouts definieren, um Blockaden zu verhindern

ePN: embedded Perl Nagios

Normalerweise forkt Nagios einen
eigenen Prozess aus und oeffnet
per popen() ein Programm. Im
Fall von PerlPlugins wird hier zu
erst der PerlInterpreter gestartet,
welcher wiederum das eigentliche
PluginProgramm laedt, kompiliert
und schliesslich ausfuehrt.

Mit ePN fuehrt der NagiosProzess
ueber einen eingebetteten Perl
Interpreter selbst das PerlPlugin
Programm aus. Zusaetzlich wird
der Bytecode des Programms und
verwendeter Module gecachet.

Plugin

NagiosProzess

ePN

fork();
popen();

Perl

Plugin

ePN: embedded Perl Nagios

● ePN muss von Nagios unterstuetzt werden
– Unterstuetzung mit uebersetzen...

● ./configure enableembeddedperl 
withperlcache ...

– ... und in der Konfiguration aktivieren
● “enable_embedded_perl” zur Aktivierung
● “use_embedded_perl_implicitly” zur 

Regelung impliziter Verwendung von ePN

– Unterstuetzung bei den meisten Distributionen 
enthalten

ePN: embedded Perl Nagios

● Steuerung expliziter (Nicht)Verwendung von 
ePN ueber optionale Kommentare

– Steuerkommentar muss in den ersten 10 Zeilen 
des Plugins festgelegt sein

– Programm mit ePN starten: # nagios: +epn
● fuehrt das Programm mit ePN aus, sofern 

implizit “normale” Ausfuehrung gewuenscht 

– Programm ohne ePN starten: # nagios: epn
●  erzwingt “normale” Ausfuehrung, sofern 

implizit Ausfuehrung mit ePN gewuenscht

Ein einfaches Beispiel

# Dieses Plugin kann mit ePN ausgefuehrt
# werden.
# nagios: +epn

use strict;
use warnings;
use Getopt::Long;

sub main(){
  # Befehlszeile parsen, WarnSchwelle in
  # $warn und kritische Schwelle in $crit
  # speichern.
  new Getopt::Long::Parser(config =>
      ['no_ignore_case', 'bundling'])
    >getoptions(
        'warning|w=i'  => \my $warn,
        'critical|c=i' => \my $crit
      )
      || do{
          print("UNKNOWN : Unable to parse"
            . " arguments!\n");
          exit(3);
        }

Der Kommentar # nagios: +epn
deklariert das Plugin explizit als
ePNkompatibel.

Wir verwenden Getopt::Long,
dies wird fuer Plugins empfohlen.
Huebscher als die uebliche API ist
die hier verwendete OOAPI.

Definiert werden die Befehlszeilen
Parameter w bzw. warning
sowie c bzw. critical, die
einen ganzzahligen Wert erwarten.
Die Benutzung von w und c fuer
Schwellwerte wird empfohlen.

Ein einfaches Beispiel

  # ACPIThermalzone 0 oeffnen...
  open(my $fd, '<', '/proc/acpi'
    . '/thermal_zone/TZ00/temperature')
    || do{
        print("UNKNOWN : Unable to open"
          . " ACPI thermal zone 0!\n");
        exit(3);
      };
  # ... und Wert sowie Einheit in $temp
  #  und $unit speichern.
  my($temp, $unit) = map{
      /^temperature:\s+(\d+)\s*(.*)$/
      ? ($1, $2)
      : ()
     } <$fd>;
  close($fd);

Ueber das procfs wird die Tem
peratur der ersten “Thermal Zone”
ausgelesen. Kann die Datei nicht
geoeffnet werden, wird eine
UNKNOWNMeldung erzeugt.

Der Temperaturwert wird in
$temp gespeichert, die Einheit
(z.B. “C” fuer Grad Celsius) wird in
$unit gespeichert.
Falls der Regex nicht greift (da der
gesuchte Inhalt nicht vorkomme),
werden beide Einheiten undef 
sein.

Ein einfaches Beispiel

  # Auf gueltigen Wert pruefen und mit kri
  # tischer und WarnSchwelle vergleichen.
  if(!defined($temp)){
    print("UNKNOWN : Unable to collect"
      . " temperature!\n");
    exit(3);
  }
  elsif(defined($crit) && $temp > $crit){
    printf("CRITICAL : Temperature %d %s"
      . " (> %d)\n", $temp, $unit, $crit);
    exit(2);
  }
  elsif(defined($warn) && $temp > $warn){
    printf("WARNING : Temperature %d %s"
      . " (> %d)\n", $temp, $unit, $warn);
    exit(1);
  }
  else{
    printf("OK : Temperature %d %s\n",
      $temp, $unit);
    exit(0);
  }
}

main();

Pruefe den Inhalt von $temp
zunaechst auf Gueltigkeit,
vergleiche anschliessend den Wert
mit den Schwellwerten in $warn
und $crit, sofern definiert.

Erzeuge anschliessend Status
meldungen unter Angabe der
Temperatur sowie ggf. des ueber
schrittenen Schwellwertes.

Der ExitCode teilt dem Monitoring
System den Status des Checks mit,
0 fuer OK, 1 fuer WARNING, 2 fuer
CRITICAL und 3 fuer UNKNOWN.

Schwellen und Bereiche

● Entwickler sind angehalten, ein einheitliches 
Format fuer Schwellen und Bereiche zu 
implementieren: “[@]Anfang:Ende”

– ohne @: Alarm falls Wert ausserhalb,

– mit @: Alarm falls Wert innerhalb Anfang:Ende

– Anfang muss kleinergleich Ende sein

– ohne obere Schwelle: “[@]Anfang:”

– ohne untere Schwelle: “[@]~:Ende”

– wird nur eine Zahl angegeben, ist es Ende; es 
wird Anfang == 0 angenommen

Schwellen und Bereiche

● Beispiele:
– check_foo w '10' c '20'

● CRITICAL, falls Wert groesser als 20 oder 
kleiner als 0 ist

● WARNING, falls Wert groesser als 10 oder 
kleiner als 0 ist

– check_foo w '10:20' c '5:25'
● CRITICAL, falls Wert groesser als 25 oder 

kleiner als 5 ist
● WARNING, falls Wert groesser als 20 oder 

kleiner als 10 ist

Schwellen und Bereiche

● Beispiele:
– check_foo w '5:10' c '~:20'

● CRITICAL, falls Wert groesser als 20 ist
● WARNING, falls Wert groesser als 10 oder 

kleiner als 5 ist

– check_foo w '@10:15' c '5:'
● CRITICAL, falls Wert kleiner als 5 ist
● WARNING, falls Wert groessergleich 10 und 

kleinergleich 15 ist

Schwellwertangaben zerlegen

# Zerlege eine ThresholdRange in Typ
# (innerhalb/ausserhalb des Bereichs) sowie
# untere und obere Schwelle, sofern
# definiert.
sub parseThreshold($){

  my($range) = @_;

  # Zerlege Schwellwerte/Bereiche per Regex.
  my($type, $min, $max) = ($range =~ /^
        (@|)                       # Type
        (?:(?\d+(?:\.\d+)?|~):)?  # Min
        (?\d+(?:\.\d+)?)?         # Max
        $/x);

  # Gebe undef zurueck, falls Parsen nicht
  # erfolgreich (durch ungueltiges Format).
  return(undef) unless(defined($type));

  # $type ist 0 fuer Alarm falls ausserhalb
  # des Bereichs und 1 fuer Alarm falls
  # innerhalb.
  $type = $type ne '' ? 1 : 0;

Die Funktion parseThreshold
dient der Zerlegung einer
Threshold wie zuvor beschrieben
in fuer uns verwertbare Elemente.

Die primaere Zerlegung findet
ueber einen regulaeren Ausdruck
statt.

Falls der Typ undef ist, hat der
Ausdruck nicht gepasst, d.h. das
Format war ungueltig.
Sonst wird er 0 fuer Alarm falls
ausserhalb und 1 falls innerhalb
des Bereichs.

Schwellwertangaben zerlegen

  # $min/$max soll undef sein falls keine
  # untere/obere Schwelle definiert.
  $min = defined($min) 
    ? ($min eq '~' ? undef : $min)
    : 0;

  # Pruefe, ob $min > $max ist...
  if(defined($min) && defined($max)
      && $min > $max){
    # ... und vertausche ggf. $min und $max.
    ($min, $max) = ($max, $min);
  }

  return($type, $min, $max);

}

Die zurueckgegebenen Werte in
$min und $max sind undef, falls
keine untere bzw. obere Schwelle
in der Schwellenangabe enthalten
war.
$max ist durch den Regex bereits
korrekt gesetzt, fuer $min
muessen wir hier noch nachhelfen.

Falls $min > $max ist, ist die
Schwellenangabe eigentlich
ungueltig, aber wir tauschen hier
einfach die Werte der Variablen
mit einander.

Performance-Daten

● Ausgabe der Mess und Grenzwerte in einem 
einheitlichen Format

– Weiterverarbeitung durch Plugins und 
Zusatzprogramme

● ermoeglichen Kapazitaets/Ressourcenplanung
– Wie haben sich Werte bei bestimmten 

Ereignissen in der Vergangenheit entwickelt?

– Wie ist werden sich die Werte voraussichtlich 
entwickeln und wann ist mit einem 
Ueberschreiten der Grenzwerte zu rechnen?

Performance-Daten

● haeufigste Anwendungen: RoundRobin
Datenbanken, Graphing und Trending

– meist manuelle Ressourcenplanung ohne 
aufwendige Statistiken

– zahlreiche Tools und Plugins, z.B.
● nagiosgraph
● BrainyPDM
● PNP4Nagios
● APAN

– oft Teil alternativer NagiosGUIs, z.B. Centreon

Performance-Daten

● Formatdefinition fuer PerformanceDaten: 
Name=Wert[Einheit];[Warn];[Krit];[Min];[Max]

– PerformanceDaten werden an die regulaere 
Ausgabe des Plugins angehaengt, getrennt 
durch ein Pipe (“|”)

– das Plugin kann PerformanceDaten zu mehreren 
Werten im o.g. Format auf einmal ausgeben, 
getrennt durch Leerzeichen

– Beispielausgabe: Temp OK: 49 C |temp=49C

Performance-Daten

● Formatdefinition fuer PerformanceDaten: 
Name=Wert[Einheit];[Warn];[Krit];[Min];[Max]

– Name darf beliebige Zeichen enthalten, sollten 
jedoch Leerzeichen, ein GleichZeichen (“=”) 
oder ein SingleQuote (“ ' ”) enthalten sein, 
muss er in SingleQuotes gesetzt werden

– SingleQuotes in Name muessen durch zwei 
SingleQuotes dargestellt werden

– alleinstehende, am Ende anhaengende 
Semikolons (“;”) duerfen weggelassen werden

Performance-Daten

● Formatdefinition fuer PerformanceDaten: 
Name=Wert[Einheit];[Warn];[Krit];[Min];[Max]

– Einheit darf offiziell nur eine der folgenden sein 
(die meisten Programme koenne jedoch mit 
beliebigen Einheiten arbeiten):

● keine Einheit: einfache Nummer/Anzahl
● B: Bytes (auch KB, MB, GB, TB, ...)
● s: Sekunden (auch us, ms, ...)
● %: Prozent (dann Default fuer Min/Max: 0/100)
● c: Zaehler (monoton steigender Wert, z.B. 

Uptime in Ticks oder Sekunden)

Performance-Daten

● Formatdefinition fuer PerformanceDaten: 
Name=Wert[Einheit];[Warn];[Krit];[Min];[Max]

– Warn und Krit sind optionale Schwellwerte fuer 
den  Warning bzw. CriticalAlarm, angegeben 
im zuvor besprochenen Format fuer 
Schwellwerte und Bereiche

– Min und Max sind die optionalen niedrigsten bzw. 
hoechsten Werte, die Wert annehmen kann

– Warn, Krit, Min und Max werden ohne Einheit 
angeben

Screenshot: PNP4Nagios

Komplexeres Beispiel

  # $min/$max soll undef sein falls keine
  # untere/obere Schwelle definiert.
  $min = defined($min) 
    ? ($min eq '~' ? undef : $min)
    : 0;

  # Pruefe, ob $min > $max ist...
  if(defined($min) && defined($max)
      && $min > $max){
    # ... und vertausche ggf. $min und $max.
    ($min, $max) = ($max, $min);
  }

  return($type, $min, $max);

}

Dieses Plugin soll Verzeichnisse
(Mountpoints) hinsichtlich ihres
verbleibenden Speicherplatzes
ueberwachen. Es ist explizit ePN
kompatibel.

Das CPANModul Filesys::Df
wird verwendet, um die Belegung
von Speicherplatz und Inodes zu
beziehen.

Wir verwenden auch hier wieder
Getopt::Long, um Befehlszei
lenargumente zu parsen.

# Dieses Plugin kann mit ePN ausgefuehrt
# werden.
# nagios: +epn

use strict;
use warnings;

use Filesys::Df;
use Getopt::Long;

sub main(){
  # Befehlszeile parsen, WarnSchwelle in
  # $warn und kritische Schwelle in
  # $crit speichern.
  new Getopt::Long::Parser(config =>
      ['no_ignore_case', 'bundling'])
    >getoptions(
        'warning|w=s'  => \my $warn,
        'critical|c=s' => \my $crit,
        'dir|d=s'      => \my $dir
      )
      || do{
          print("UNKNOWN : Unable to parse"
            . " arguments!\n");
          exit(3);
        };

Komplexeres Beispiel

$dir enthaelt das zu ueberwach
ende Verzeichnis. Es muss
definiert, d.h. ueber die Befehls
zeile uebergeben worden sein und
explizit existieren.

$status enthaelt den ExitCode,
mit dem das Plugin sich beendet.
Default ist 0 fuer “OK”.

df() erzeugt einen anonymen
Hash, der die Daten es ueber
gebenen Verzeichnisses enthaelt,
Speicherplatzangaben moechten
wir in MiB auswerten.

# Dieses Plugin kann mit ePN ausgefuehrt
# werden.
# nagios: +epn

use strict;
use warnings;

use Getopt::Long;
use Filesys::Df;

sub main(){
  # Befehlszeile parsen, WarnSchwelle in
  # $warn und kritische Schwelle in
  # $crit speichern.
  new Getopt::Long::Parser(config =>
      ['no_ignore_case', 'bundling'])
    >getoptions (
        'warning|w=s'  => \my $warn,
        'critical|c=s' => \my $crit,
        'dir|d=s'      => \my $dir
      )
      || do{
          print ("UNKNOWN : Unable to parse"
            . " arguments!\n");
          exit (3);
        };

  # $dir muss definiert...
  unless(defined($dir) && $dir ne ''){
    print("UNKNOWN: No directory given to"
      . " monitor.\n");
    exit(3);
  }
  # ... und ein existierendes Verzeichnis
  # sein.
  unless(d $dir){
    printf("UNKNOWN: Directory not found:"
      . " %s\n", $dir);
    exit(3);
  }

  # DefaultExitcode.
  my $status = 0;

  # Wir interessieren uns fuer die Angabe
  # in MiB.
  my $df = df($dir, 1024*1024);

Komplexeres Beispiel

Die Funktion isOutOfRange()
prueft, ob ein Wert, hier die pro
zentuale Speicherbelegung, inner
halb einer uebergebenen Schwell
wertdefinition liegt. Sie liefert einen
wahren oder falschen Wert bzw.
undef, falls die Schwellwerte
nicht geparset werden konnten.

In einem ifelsifelseBaum
werden die zurueckgegebenen
Werte entsprechend ihrer Dring
lichkeit abgearbeitet, Ausgaben
erzeugt und der ExitCode fest
gelegt.

  # $dir muss definiert...
  unless(defined($dir) && $dir ne ''){
    print("UNKNOWN: No directory given to"
      . " monitor.\n");
    exit(3);
  }
  # ... und ein existierendes Verzeichnis
  # sein.
  unless(d $dir){
    printf("UNKNOWN: Directory not found:"
      . " %s\n", $dir);
    exit(3);
  }

  # DefaultExitcode.
  my $status = 0;

  # Wir interessieren uns fuer die Angabe
  # in MiB.
  my $df = df($dir, 1024*1024);

  # Speichere bool'sche Werte, ob eine der
  # Schwellen verletzt ist (oder undef,
  # falls Parsing fehlschlaegt).
  my $isCrit 
    = isOutOfRange($df>{per}, $crit);
  my $isWarn 
    = isOutOfRange($df>{per}, $warn);
  
  # Werte Schwellwertverletzungen aus.
  if(!defined($isCrit)
      || !defined($isWarn)){
    print("UNKNOWN: Bad threshold format");
    $status = 3;
  }
  elsif($isCrit){
    printf("CRITICAL: Disk %s usage %d%%"
      . " (out of threshold %s)",
      $dir, $df>{per}, $crit);
    $status = 2;
  }
  elsif($isWarn){
    printf("WARNING: Disk %s usage %d%%"
      . " (out of threshold %s)", 
      $dir, $df>{per}, $warn);
    $status = 1;
  }

Nun werden noch Performance
Daten erzeugt und ausgegeben,
hier zur Platzbelegung in MiB sowie
zu den belegten Inodes, sofern das
Dateisystem diese definiert.

Hier wird neben dem Messwert nur
der Mindestwert von 0 sowie der
Maximalwert (MiB bzw. Inodes ins
gesamt) ausgegeben. Warning
und CriticalSchwellen sind ja fuer
diese Werte nicht definiert.

Schliesslich ein Newline ausgeben
und mit ExitCode beenden.

  else{
    printf("OK: Disk %s usage %d%%", 
      $dir, $df>{per});
  }
  
  # Gebe PerformanceDaten zum belegten
  # Speicherplatz sowie, sofern verfuegbar,
  # den belegten Inodes aus.
  printf(" | 'Space Usage'=%dMB;;;0;%d",
    $df>{used}, $df>{blocks});
  if(exists($df>{files})){
    printf(" 'Inode Usage'=%d;;;0;%d",
      $df>{fused}, $df>{files});
  }
  
  # Schliesse die Zeile mit Newline ab und
  # beende mit dem oben festgelegten
  # Exitcode.
  print("\n");
  exit($status);

}

Die Funktion isOutOfRange().

Wenn keine Schwelle definiert ist,
kann der Wert nicht ausserhalb
liegen, gebe einen FalschWert
zurueck.

Parse die Schwelle und gebe bei
einem Fehler undef zurueck.

$oor speichert das Resultat der
Vergleiche, ob der Wert ausser
halb der unteren/oberen Schwelle
liegt, sofern definiert.

.# Pruefe, ob $value ausserhalb von $range
# liegt.
sub isOutOfRange($$){
  my($value, $range) = @_;
  
  # Wenn keine Schwelle definiert ist,
  # liegt der Wert nicht ausserhalb.
  return(0) unless(defined($range));
  
  # Die Schwelle parsen und undef zurueck
  # geben, falls das Parsen fehlschlaegt.
  my($type, $min, $max)
    = parseThreshold($range);
  return(undef) unless(defined($type));
  
  # $oor erhaelt einen wahren bool'schen
  # Wert, falls der Mindest oder Hoechst
  # wert der Schwelle, sofern definiert,
  # unter/ueberschritten ist.
  my $oor 
    = ((defined($min) && $value < $min) 
    || (defined($max) && $value > $max));

Komplexeres Beispiel

Falls die Schwelle invertiert ist
(vorangestelltes “@”), muss das
Resultat invertiert zurueckgegeben
werden.

Die Funktion parseThreshold()
entspricht der vorab besprochenen
Funktion mit gleichem Namen.

  # $oor zurueckgeben, falls die Schwelle
  # vom Typ "Alarm falls innerhalb" ist,
  # negiert.
  return($type ? !$oor : $oor);
}

sub parseThreshold($){
  ...
}

main();
1;

Vielen Dank
fuer Eure Aufmerksamkeit



Seite 38

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 8. Februar 2010

Links

● Nagios
– http://www.nagios.org/

● Icinga
– http://www.icinga.org/

● Centreon
– http://www.centreon.com/

● PNP4Nagios
– http://www.pnp4nagios.org/

Links

● BrainyPDM
– http://www.brainypdm.org/

● APAN
– http://apan.sf.net/

● nagiosgraph
– http://nagiosgraph.sf.net/

● NetSNMP
– http://www.netsnmp.org/

Links

● nsclient++ / nscp
– http://nsclient.org/ 

● nsca und nrpe
– http://www.nagios.org/



#!/usr/bin/perl

# Dieses Plugin kann mit ePN ausgefuehrt werden.
# nagios: +epn

use strict;
use warnings;

use Getopt::Long;

sub main(){
	# Befehlszeile parsen, Warn-Schwelle in $warn und kritische Schwelle in
	# $crit speichern.
	new Getopt::Long::Parser(config => ['no_ignore_case', 'bundling'])
  	->getoptions('warning|w=i' => \my $warn, 'critical|c=i' => \my $crit)
	  	|| do{
					print("UNKNOWN : Unable to parse arguments!\n");
					exit(3);
				};
	
	# ACPI-Thermalzone 0 oeffnen...
	open(my $fd, '<', '/proc/acpi/thermal_zone/TZ00/temperature')
  	|| do{
				print("UNKNOWN : Unable to open ACPI thermal zone 0!\n");
				exit(3);
			};
	# ... und Wert sowie Einheit in $temp und $unit speichern.
	my($temp, $unit) = map{/^temperature:\s+(\d+)\s*(.*)$/ ? ($1, $2) : ()} <$fd>;
	close($fd);
	
	# Auf gueltigen Wert pruefen und mit kritischer und Warn-Schwelle vergleichen.
	if(!defined($temp)){
		print("UNKNOWN : Unable to collect temperature!\n");
		exit(3);
	}
	elsif(defined($crit) && $temp > $crit){
		printf("CRITICAL : Temperature %d %s (> %d)\n", $temp, $unit, $crit);
		exit(2);
	}
	elsif(defined($warn) && $temp > $warn){
		printf("WARNING : Temperature %d %s (> %d)\n", $temp, $unit, $warn);
		exit(1);
	}
	else{
		printf("OK : Temperature %d %s\n", $temp, $unit);
		exit(0);
	}
}

main();
1;


#!/usr/bin/perl

# Dieses Plugin kann mit ePN ausgefuehrt werden.
# nagios: +epn

use strict;
use warnings;

use Getopt::Long;
use Filesys::Df;

sub main(){
	# Befehlszeile parsen, Warn-Schwelle in $warn und kritische Schwelle in
	# $crit speichern.
	new Getopt::Long::Parser(config => ['no_ignore_case', 'bundling'])
  	->getoptions('warning|w=s' => \my $warn, 'critical|c=s' => \my $crit,
  	             'dir|d=s'     => \my $dir)
	  	|| do{
					print("UNKNOWN : Unable to parse arguments!\n");
					exit(3);
				};
	
	# $dir muss definiert...
	unless(defined($dir) && $dir ne ''){
		print("UNKNOWN: No directory given to monitor.\n");
		exit(3);
	}
	# ... und ein existierendes Verzeichnis sein.
	unless(-d $dir){
		printf("UNKNOWN: Directory not found: %s\n", $dir);
		exit(3);
	}
	
	# Default-Exitcode.
	my $status = 0;
	
	# Wir interessieren uns fuer die Angabe in MiB.
	my $df = df($dir, 1024*1024);
	
	# Speichere bool'sche Werte, ob eine der Schwellen verletzt ist
	# (oder undef, falls Parsing fehlschlaegt).
	my $isCrit = isOutOfRange($df->{per}, $crit);
	my $isWarn = isOutOfRange($df->{per}, $warn);
	
	# Werte Schwellwertverletzungen aus.
	if(!defined($isCrit) || !defined($isWarn)){
		print("UNKNOWN: Bad threshold format");
		$status = 3;
	}
	elsif($isCrit){
		printf("CRITICAL: Disk %s usage %d%% (out of threshold %s)", $dir, $df->{per}, $crit);
		$status = 2;
	}
	elsif($isWarn){
		printf("WARNING: Disk %s usage %d%% (out of threshold %s)", $dir, $df->{per}, $warn);
		$status = 1;
	}
	else{
		printf("OK: Disk %s usage %d%%", $dir, $df->{per});
	}
	
	# Gebe Performance-Daten zum belegten Speicherplatz sowie, sofern verfuegbar, den
	# belegten Inodes aus.
	printf(" | 'Space Usage'=%dMB;;;0;%d", $df->{used}, $df->{blocks});
	if(exists($df->{files})){
		printf(" 'Inode Usage'=%d;;;0;%d", $df->{fused}, $df->{files});
	}
	
	# Schliesse die Zeile mit Newline ab und beende mit dem oben festgelegten Exitcode.
	print("\n");
	exit($status);
}

# Pruefe, ob $value ausserhalb von $range liegt.
sub isOutOfRange($$){
	my($value, $range) = @_;
	
	# Wenn keine Schwelle definiert ist, liegt der Wert nicht ausserhalb.
	return(0) unless(defined($range));
	
	# Die Schwelle parsen und undef zurueckgeben, falls das Parsen fehlschlaegt.
	my($type, $min, $max) = parseThreshold($range);
	return(undef) unless(defined($type));
	
	# $oor erhaelt einen wahren bool'schen Wert, falls der Mindest- oder
	# Hoechstwert der Schwelle, sofern definiert, unter-/ueberschritten ist.
	my $oor = ((defined($min) && $value < $min) || (defined($max) && $value > $max));
	
	# $oor zurueckgeben, falls die Schwelle vom Typ "Alarm falls innerhalb" ist,
	# negiert.
	return($type ? !$oor : $oor);
}

# Zerlege eine Threshold-Range in Typ (innerhalb/ausserhalb des Bereichs) sowie
# untere und obere Schwelle, sofern definiert.
sub parseThreshold($){
	my($range) = @_;
	
	# Zerlege Schwellwerte/Bereiche per Regex in Type, Min und Max.
	my($type, $min, $max) = ($range =~ /^(@|)(?:(-?\d+(?:\.\d+)?|~):)?(-?\d+(?:\.\d+)?)?$/);
	
	# Gebe undef zurueck, falls Parsen nicht erfolgreich (durch ungueltiges
	# Format).
	return(undef) unless(defined($type));
	
	# $type ist 0 fuer Alarm falls ausserhalb des Bereichs und 1 fuer Alarm falls
	# innerhalb.
	$type = $type ne '' ? 1 : 0;

	# $min/$max soll undef sein falls keine untere/obere Schwelle definiert.
	$min = defined($min) ? ($min eq '~' ? undef : $min) : 0;
	
	# Pruefe, ob $min > $max ist und vertausche ggf. $min und $max.
	($min, $max) = ($max, $min) if(defined($min) && defined($max) && $min > $max);
	
	return($type, $min, $max);
}

main();
1;




#!/usr/bin/perl

use strict;
use warnings;

# Zerlege eine Threshold-Range in Typ (innerhalb/ausserhalb des Bereichs) sowie
# untere und obere Schwelle, sofern definiert.
sub parseThreshold($){
	my($range) = @_;
	
	# Zerlege Schwellwerte/Bereiche per Regex in Type, Min und Max.
	my($type, $min, $max) = ($range =~ /^(@|)(?:(-?\d+(?:\.\d+)?|~):)?(-?\d+(?:\.\d+)?)?$/);
	
	# Gebe undef zurueck, falls Parsen nicht erfolgreich (durch ungueltiges
	# Format).
	return(undef) unless(defined($type));
	
	# $type ist 0 fuer Alarm falls ausserhalb des Bereichs und 1 fuer Alarm falls
	# innerhalb.
	$type = $type ne '' ? 1 : 0;

	# $min/$max soll undef sein falls keine untere/obere Schwelle definiert.
	$min = defined($min) ? ($min eq '~' ? undef : $min) : 0;
	
	# Pruefe, ob $min > $max ist und vertausche ggf. $min und $max.
	($min, $max) = ($max, $min) if(defined($min) && defined($max) && $min > $max);
	
	return($type, $min, $max);
}

print(join(', ', map{defined($_) ? $_ : 'undef'} parseThreshold($ARGV[0])) . "\n");


